

# Combinatorics on Words formalized Graph Lemma

Štěpán Holub  
Štěpán Starosta

October 13, 2025

Funded by the Czech Science Foundation grant GAČR 20-20621S.

# Contents

<b>1</b>	<b>Glued codes</b>	<b>2</b>
1.1	Lists that do not end with a fixed letter . . . . .	2
1.2	Glue a list element with its successors/predecessors . . . . .	3
1.3	Generators with glued element . . . . .	5
1.4	Bounded gluing . . . . .	7
1.4.1	Gluing on binary alphabet . . . . .	7
1.5	Code with glued element . . . . .	8
1.6	Gluing is primitivity preserving . . . . .	10
1.6.1	Gluing on binary alphabet . . . . .	12
<b>2</b>	<b>Graph Lemma</b>	<b>13</b>
2.1	Graph lemma . . . . .	13
2.2	Binary code . . . . .	14
	<b>References</b>	<b>16</b>

```

theory Glued-Codes
  imports Combinatorics-Words.Submonoids
begin

```

# Chapter 1

## Glued codes

### 1.1 Lists that do not end with a fixed letter

**lemma** *append-last-neq*:

$us = \varepsilon \vee \text{last } us \neq w \implies vs = \varepsilon \vee \text{last } vs \neq w \implies us \cdot vs = \varepsilon \vee \text{last } (us \cdot vs) \neq w$   
**by** (*auto simp only: last-append split: if-split*)

**lemma** *last-neq-induct* [*consumes 1, case-names emp hd-eq hd-neq*]:

**assumes** *invariant*:  $us = \varepsilon \vee \text{last } us \neq w$   
**and** *emp*:  $P \ \varepsilon$   
**and** *hd-eq*:  $\bigwedge us. us \neq \varepsilon \implies \text{last } us \neq w \implies P \ us \implies P \ (w \# us)$   
**and** *hd-neq*:  $\bigwedge u \ us. u \neq w \implies us = \varepsilon \vee \text{last } us \neq w \implies P \ us \implies P \ (u \# us)$   
**shows**  $P \ us$   
**using** *invariant proof* (*induction us*)  
**case** (*Cons u us*)  
**have** *inv*:  $us = \varepsilon \vee \text{last } us \neq w$   
**using** *Cons.prem*s **by** (*intro disjI*) *simp*  
**show**  $P \ (u \# us)$   
**proof** (*cases*)  
**assume**  $u = w$   
**have** \*:  $us \neq \varepsilon$  **and**  $\text{last } us \neq w$   
**using** *Cons.prem*s **unfolding**  $\langle u = w \rangle$  **by** *auto*  
**then show**  $P \ (u \# us)$  **unfolding**  $\langle u = w \rangle$  **using** *Cons.IH*[*OF inv*] **by** (*fact hd-eq*)  
**qed** (*use inv Cons.IH*[*OF inv*] **in**  $\langle \text{fact hd-neq} \rangle$ )  
**qed** (*rule*  $\langle P \ \varepsilon \rangle$ )

**lemma** *last-neq-blockE*:

**assumes** *last-neq*:  $us \neq \varepsilon$  **and**  $\text{last } us \neq w$   
**obtains**  $k \ u \ us'$  **where**  $u \neq w$  **and**  $us' = \varepsilon \vee \text{last } us' \neq w$  **and**  $[w]^{\textcircled{a}} k \cdot u \# us' = us$   
**using** *disjI2*[*OF*  $\langle \text{last } us \neq w \rangle$ ]  $\langle us \neq \varepsilon \rangle$  **proof** (*induction us rule: last-neq-induct*)  
**case** (*hd-eq us*)

```

from  $\langle us \neq \varepsilon \rangle$  show ?case
  by (rule hd-eq.IH[rotated]) (intro hd-eq.prem(1)[of - - Suc -], assumption+,
simp)
next
  case (hd-neq u us)
    from hd-neq.hyps show ?case
    by (rule hd-neq.prem(1)[of - - 0]) simp
qed blast

lemma last-neq-block-induct [consumes 1, case-names emp block]:
  assumes last-neq:  $us = \varepsilon \vee \text{last } us \neq w$ 
  and emp:  $P \ \varepsilon$ 
  and block:  $\bigwedge k \ u \ us. u \neq w \implies us = \varepsilon \vee \text{last } us \neq w \implies P \ us \implies P \ ([w]^\text{@} k \cdot (u \# us))$ 
  shows  $P \ us$ 
using last-neq proof (induction us rule: ssuf-induct)
  case (ssuf us)
    show ?case proof (cases us =  $\varepsilon$ )
      assume  $us \neq \varepsilon$ 
      obtain  $k \ u \ us'$  where  $u \neq w$  and  $us' = \varepsilon \vee \text{last } us' \neq w$  and  $[w]^\text{@} k \cdot u \# us' = us$ 
      using  $\langle us \neq \varepsilon \rangle \ \langle us = \varepsilon \vee \text{last } us \neq w \rangle$  by (elim last-neq-blockE) (simp add:  $\langle us \neq \varepsilon \rangle$ )
      have  $us' <_s us$  and  $us' = \varepsilon \vee \text{last } us' \neq w$ 
      using  $\langle us = \varepsilon \vee \text{last } us \neq w \rangle$  by (auto simp flip:  $\langle [w]^\text{@} k \cdot u \# us' = us \rangle$ )
      from  $\langle u \neq w \rangle \ \langle us' = \varepsilon \vee \text{last } us' \neq w \rangle$  ssuf.IH[OF this]
      show  $P \ us$  unfolding  $\langle [w]^\text{@} k \cdot u \# us' = us \rangle$  [symmetric] by (fact block)
    qed (simp only: emp)
qed

```

## 1.2 Glue a list element with its successors/predecessors

```

function glue :: 'a list  $\Rightarrow$  'a list list  $\Rightarrow$  'a list list where
  glue-emp: glue w  $\varepsilon = \varepsilon$  |
  glue-Cons: glue w (u # us) =
    (let glue-tl = glue w us in
     if u = w then (u · hd glue-tl) # tl glue-tl
     else u # glue-tl)
  unfolding prod-eq-iff prod.sel by (cases rule: list.exhaust[of snd -]) blast+
  termination by (relation measure (length  $\circ$  snd)) simp-all

lemma no-gluing:  $w \notin \text{set } us \implies \text{glue } w \ us = us$ 
  by (induction us) auto

lemma glue-nemp [simp, intro!]:  $us \neq \varepsilon \implies \text{glue } w \ us \neq \varepsilon$ 
  by (elim hd-tlE) (auto simp only: glue.simps Let-def split!: if-split)

```

**lemma** *glue-is-emp-iff* [simp]:  $glue\ w\ us = \varepsilon \longleftrightarrow us = \varepsilon$   
**using** *glue-nemp glue-emp* **by** *blast*

**lemma** *len-glue*:  $us = \varepsilon \vee last\ us \neq w \implies |glue\ w\ us| + count-list\ us\ w = |us|$   
**by** (*induction rule: last-neq-induct*) (*auto simp add: Let-def*)

**lemma** *len-glue-le*: **assumes**  $us = \varepsilon \vee last\ us \neq w$  **shows**  $|glue\ w\ us| \leq |us|$   
**using** *len-glue[OF assms]* **unfolding** *nat-le-iff-add eq-commute[of |us|]* **by** *blast*

**lemma** *len-glue-less* []:  $us = \varepsilon \vee last\ us \neq w \implies w \in set\ us \implies |glue\ w\ us| < |us|$   
**by** (*simp add: count-list-gr-0-iff flip: len-glue[of us]*)

**lemma** **assumes**  $us = \varepsilon \vee last\ us \neq w$  **and**  $\varepsilon \notin set\ us$   
**shows** *emp-not-in-glue*:  $\varepsilon \notin set\ (glue\ w\ us)$   
**and** *glued-not-in-glue*:  $w \notin set\ (glue\ w\ us)$   
**unfolding** *atomize-conj* **using** *assms* **by** (*induction us rule: last-neq-induct*)  
(*auto simp: Let-def dest!: tl-set lists-hd-in-set[OF glue-nemp[of - w]]*)

**lemma** *glue-glue*:  $us = \varepsilon \vee last\ us \neq w \implies \varepsilon \notin set\ us \implies glue\ w\ (glue\ w\ us) = glue\ w\ us$   
**using** *no-gluing[OF glued-not-in-glue]*.

**lemma** *glue-block-append*: **assumes**  $u \neq w$   
**shows**  $glue\ w\ ([w]^@ k \cdot (u \# us)) = (w^@ k \cdot u) \# glue\ w\ us$   
**by** (*induction k*) (*simp-all add: <u ≠ w>*)

**lemma** *concat-glue* [simp]:  $us = \varepsilon \vee last\ us \neq w \implies concat\ (glue\ w\ us) = concat\ us$   
**by** (*induction us rule: last-neq-block-induct*) (*simp-all add: glue-block-append*)

**lemma** *glue-append*:  
 $us = \varepsilon \vee last\ us \neq w \implies glue\ w\ (us \cdot vs) = glue\ w\ us \cdot glue\ w\ vs$   
**by** (*induction us rule: last-neq-block-induct*) (*simp-all add: glue-block-append*)

**lemma** *glue-pow*:  
**assumes**  $us = \varepsilon \vee last\ us \neq w$   
**shows**  $glue\ w\ (us^@ k) = (glue\ w\ us)^@ k$   
**by** (*induction k*) (*simp-all add: assms glue-append*)

**lemma** *glue-in-lists-hull* [intro]:  
 $us = \varepsilon \vee last\ us \neq w \implies us \in lists\ G \implies glue\ w\ us \in lists\ \langle G \rangle$   
**by** (*induction rule: last-neq-induct*) (*simp-all add: Let-def tl-in-lists prod-cl gen-in*)

— Gluing from the right (gluing a letter with its predecessor)

**function** *gluer* :: 'a list  $\Rightarrow$  'a list list  $\Rightarrow$  'a list list **where**  
*gluer-emp*:  $gluer\ w\ \varepsilon = \varepsilon$  |  
*gluer-Cons*:  $gluer\ w\ (u \# us) =$   
(*let* *gluer-butlast* =  $gluer\ w\ (butlast\ (u \# us))$  *in*  
*if*  $last\ (u \# us) = w$  *then*  $(butlast\ gluer-butlast) \cdot [last\ gluer-butlast \cdot last\ (u$

```

# us)]
  else gluer-butlast · [last (u # us)]))
unfolding prod-eq-iff prod.sel by (cases rule: list.exhaust[of snd -]) blast+
termination by (relation measure (length ∘ snd)) simp-all

```

```

lemma gluer-nemp-def: assumes us ≠ ε
shows gluer w us =
  (let gluer-butlast = gluer w (butlast us) in
    if last us = w then (butlast gluer-butlast) · [last gluer-butlast · last us]
    else gluer-butlast · [last us])
using gluer-Cons[of w hd us tl us] unfolding hd-Cons-tl[OF ⟨us ≠ ε⟩].

```

```

lemma gluer-nemp: assumes us ≠ ε shows gluer w us ≠ ε
unfolding gluer-nemp-def[OF ⟨us ≠ ε⟩]
by (simp only: Let-def split!: if-split)

```

```

lemma hd-neq-induct [consumes 1, case-names emp snoc-eq snoc-neq]:
assumes invariant: us = ε ∨ hd us ≠ w
and emp: P ε
and snoc-eq:  $\bigwedge us. us \neq \varepsilon \implies hd\ us \neq w \implies P\ us \implies P\ (us \cdot [w])$ 
and snoc-neq:  $\bigwedge u\ us. u \neq w \implies us = \varepsilon \vee hd\ us \neq w \implies P\ us \implies P\ (us \cdot [u])$ 
shows P us
using last-neq-induct[where P=λx. P (rev x) for P, reversed, unfolded rev-rev-ident,
OF assms].

```

```

lemma gluer-rev [reversal-rule]: assumes us = ε ∨ last us ≠ w
shows gluer (rev w) (rev (map rev us)) = rev (map rev (glue w us))
using assms by (induction us rule: last-neq-induct)
  (simp-all add: gluer-nemp-def Let-def map-tl last-rev hd-map)

```

```

lemma glue-rev [reversal-rule]: assumes us = ε ∨ hd us ≠ w
shows glue (rev w) (rev (map rev us)) = rev (map rev (gluer w us))
using assms by (induction us rule: hd-neq-induct)
  (simp-all add: gluer-nemp-def Let-def map-tl last-rev hd-map)

```

### 1.3 Generators with glued element

The following set will turn out to be the generating set of all words whose decomposition into a generating code does not end with w

```

inductive-set glued-gens :: 'a list ⇒ 'a list set ⇒ 'a list set
for w G where
  other-gen:  $g \in G \implies g \neq w \implies g \in glued-gens\ w\ G$ 
  | glued [intro!]:  $u \in glued-gens\ w\ G \implies w \cdot u \in glued-gens\ w\ G$ 

```

```

lemma in-glued-gensI: assumes g ∈ G g ≠ w
shows w@ k · g = u ⇒ u ∈ glued-gens w G
by (induction k arbitrary: u) (auto simp: other-gen[OF ⟨g ∈ G⟩ ⟨g ≠ w⟩])

```

**lemma** *in-glued-gensE*:  
 assumes  $u \in \text{glued-gens } w \ G$   
 obtains  $k \ g$  where  $g \in G$  and  $g \neq w$  and  $w @ k \cdot g = u$   
 using *assms* **proof** (*induction*)  
 case (*glued u*)  
 show ?case **by** (*auto intro!*: *glued.IH[OF glued.prem[of - Suc -]]*)  
**qed** (*use pow-zero in blast*)

**lemma** *glued-gens-alt-def*:  $\text{glued-gens } w \ C = \{w @ k \cdot g \mid k \ g. \ g \in C \wedge g \neq w\}$   
**by** (*blast elim!*: *in-glued-gensE intro: in-glued-gensI*)

**lemma** *glued-hull-sub-hull* [*simp, intro!*]:  $w \in G \implies \langle \text{glued-gens } w \ G \rangle \subseteq \langle G \rangle$   
**by** (*rule hull-mono'*) (*auto elim!*: *in-glued-gensE*)

**lemma** *glued-hull-sub-hull'*:  $w \in G \implies u \in \langle \text{glued-gens } w \ G \rangle \implies u \in \langle G \rangle$   
 using *set-mp[OF glued-hull-sub-hull]*.

**lemma** *in-glued-hullE*:  
 assumes  $w \in G$  and  $u \in \langle \text{glued-gens } w \ G \rangle$   
 obtains *us* where  $\text{concat } us = u$  and  $us \in \text{lists } G$  and  $us = \varepsilon \vee \text{last } us \neq w$   
 using  $\langle u \in \langle \text{glued-gens } w \ G \rangle \rangle$  **proof** (*induction arbitrary: thesis*)  
 case (*prod-cl v u*)  
 obtain  $k \ g$  where  $g \in G$  and  $g \neq w$  and  $\text{concat } ([w] @ k \cdot [g]) = v$   
 using  $\langle v \in \text{glued-gens } w \ G \rangle$  **by** *simp* (*elim in-glued-gensE*)  
 obtain *us* where  $u: \text{concat } us = u$  and  $us \in \text{lists } G$  and  $(us = \varepsilon \vee \text{last } us \neq w)$  **by** *fact*  
 have  $\text{concat } ([w] @ k \cdot [g] \cdot us) = v \cdot u$   
**by** (*simp flip*:  $\langle \text{concat } ([w] @ k \cdot [g]) = v \rangle \langle \text{concat } us = u \rangle$ )  
 with  $\langle (us = \varepsilon \vee \text{last } us \neq w) \rangle$  **show** *thesis*  
**by** (*elim prod-cl.prem[1], intro lists.intros*  
*append-in-lists pow-in-lists*  $\langle w \in G \rangle \langle g \in G \rangle \langle us \in \text{lists } G \rangle$   
*(auto simp:*  $\langle g \neq w \rangle$ *)*)  
**qed** (*use concat.simps(1) in blast*)

**lemma** *glue-in-lists* [*simp, intro!*]:  
 assumes  $us = \varepsilon \vee \text{last } us \neq w$   
 shows  $us \in \text{lists } G \implies \text{glue } w \ us \in \text{lists } (\text{glued-gens } w \ G)$   
 using *assms* **by** (*induction rule: last-neq-block-induct*)  
 (*auto simp: glue-block-append intro: in-glued-gensI*)

**lemma** *concat-in-glued-hull*[*intro*]:  
 $us \in \text{lists } G \implies us = \varepsilon \vee \text{last } us \neq w \implies \text{concat } us \in \langle \text{glued-gens } w \ G \rangle$   
**unfolding** *concat-glue[symmetric]* **by** (*intro concat-in-hull' glue-in-lists*)

**lemma** *glued-hull-conv*: **assumes**  $w \in G$   
**shows**  $\langle \text{glued-gens } w \ G \rangle = \{\text{concat } us \mid us. \ us \in \text{lists } G \wedge (us = \varepsilon \vee \text{last } us \neq w)\}$   
**by** (*blast elim!*: *in-glued-hullE[OF*  $\langle w \in G \rangle$ *]*)

## 1.4 Bounded gluing

**lemma** *bounded-glue-in-lists*:

**assumes**  $us = \varepsilon \vee \text{last } us \neq w$  **and**  $\neg [w]^@ n \leq_f us$   
**shows**  $us \in \text{lists } G \implies \text{glue } w \text{ } us \in \text{lists } \{w^@ k \cdot g \mid k \cdot g. g \in G \wedge g \neq w \wedge k < n\}$   
**using** *assms proof (induction us rule: last-neq-block-induct)*  
**case** (*block k u us*)  
**have**  $k < n$  **and**  $\neg [w]^@ n \leq_f us$   
**using**  $\langle \neg [w]^@ n \leq_f [w]^@ k \cdot u \# us \rangle$   
**by** (*blast intro!: not-le-imp-less, blast intro!: fac-ext-pref fac-ext-hd*)  
**then show** ?*case*  
**using**  $\langle [w]^@ k \cdot u \# us \in \text{lists } G \rangle \langle u \neq w \rangle$  **unfolding** *glue-block-append[OF*  
 $\langle u \neq w \rangle]$   
**by** (*blast intro!: block.IH del: in-listsD in-listsI*)  
**qed simp**

### 1.4.1 Gluing on binary alphabet

**lemma** *bounded-bin-glue-in-lists*: — meaning: a binary code

**assumes**  $us = \varepsilon \vee \text{last } us \neq x$   
**and**  $\neg [x]^@ n \leq_f us$   
**and**  $us \in \text{lists } \{x, y\}$   
**shows**  $\text{glue } x \text{ } us \in \text{lists } \{x^@ k \cdot y \mid k. k < n\}$   
**using** *bounded-glue-in-lists[OF assms]* **by** *blast*

**lemma** *single-bin-glue-in-lists*: — meaning: a single occurrence

**assumes**  $us = \varepsilon \vee \text{last } us \neq x$   
**and**  $\neg [x, x] \leq_f us$   
**and**  $us \in \text{lists } \{x, y\}$   
**shows**  $\text{glue } x \text{ } us \in \text{lists } \{x \cdot y, y\}$   
**using** *bounded-bin-glue-in-lists[of - - 2, simplified, OF assms]* **unfolding** *numeral-nat*  
**by** (*auto elim!: sub-lists-mono[rotated] less-SucE*)

**lemma** *count-list-single-bin-glue*:

**assumes**  $x \neq \varepsilon$  **and**  $x \neq y$   
**and**  $us = \varepsilon \vee \text{last } us \neq x$   
**and**  $us \in \text{lists } \{x, y\}$   
**and**  $\neg [x, x] \leq_f us$   
**shows**  $\text{count-list } (\text{glue } x \text{ } us) (x \cdot y) = \text{count-list } us \ x$   
**and**  $\text{count-list } (\text{glue } x \text{ } us) y + \text{count-list } us \ x = \text{count-list } us \ y$   
**using** *assms(3–5) unfolding atomize-conj pow-Suc[symmetric]*  
**proof** (*induction us rule: last-neq-block-induct*)  
**case** (*block k u us*)  
**have**  $u = y$  **using**  $\langle [x]^@ k \cdot u \# us \in \text{lists } \{x, y\} \rangle \langle u \neq x \rangle$  **by** *simp*  
**have** *IH*:  $\text{count-list } (\text{glue } x \text{ } us) (x \cdot y) = \text{count-list } us \ x \wedge$   
 $\text{count-list } (\text{glue } x \text{ } us) y + \text{count-list } us \ x = \text{count-list } us \ y$   
**using** *block.prem* **by** (*intro block.IH*) (*simp, blast intro!: fac-ext-pref fac-ext-hd*)  
**have**  $\neg [x]^@ \text{Suc } (\text{Suc } 0) \leq_f [x]^@ k \cdot u \# us$



```

    using block.premis(2) by auto
  then have  $k < \text{Suc } (\text{Suc } 0)$ 
    by (blast intro!: not-le-imp-less)
  then show ?case unfolding  $\langle u = y \rangle$  glue-block-append[OF  $\langle x \neq y \rangle$ [symmetric]]
    by (elim less-SucE less-zeroE) (simp-all add:  $\langle x \neq y \rangle$   $\langle x \neq y \rangle$ [symmetric]  $\langle x \neq \varepsilon \rangle$  IH)
qed simp

```

## 1.5 Code with glued element

```

context code
begin

```

If the original generating set is a code, then also the glued generators form a code

```

lemma glued-hull-last-dec: assumes  $w \in \mathcal{C}$  and  $u \in \langle \text{glued-gens } w \mathcal{C} \rangle$  and  $u \neq \varepsilon$ 
  shows  $\text{last } (\text{Dec } \mathcal{C} \ u) \neq w$ 
  using  $\langle u \in \langle \text{glued-gens } w \mathcal{C} \rangle \rangle$ 
  by (elim in-glued-hullE[OF  $\langle w \in \mathcal{C} \rangle$ ]) (auto simp: code-unique-dec  $\langle u \neq \varepsilon \rangle$ )

```

```

lemma in-glued-hullI [intro]:
  assumes  $u \in \langle \mathcal{C} \rangle$  and  $(u = \varepsilon \vee \text{last } (\text{Dec } \mathcal{C} \ u) \neq w)$ 
  shows  $u \in \langle \text{glued-gens } w \mathcal{C} \rangle$ 
  using concat-in-glued-hull[OF dec-in-lists[OF  $\langle u \in \langle \mathcal{C} \rangle \rangle$ , of  $w$ ]]
  by (simp add:  $\langle u \in \langle \mathcal{C} \rangle \rangle$   $\langle u = \varepsilon \vee \text{last } (\text{Dec } \mathcal{C} \ u) \neq w \rangle$ )

```

```

lemma code-glued-hull-conv: assumes  $w \in \mathcal{C}$ 
  shows  $\langle \text{glued-gens } w \mathcal{C} \rangle = \{u \in \langle \mathcal{C} \rangle. u = \varepsilon \vee \text{last } (\text{Dec } \mathcal{C} \ u) \neq w\}$ 
proof
  show  $\langle \text{glued-gens } w \mathcal{C} \rangle \subseteq \{u \in \langle \mathcal{C} \rangle. u = \varepsilon \vee \text{last } (\text{Dec } \mathcal{C} \ u) \neq w\}$ 
    using glued-hull-sub-hull[OF  $\langle w \in \mathcal{C} \rangle$ ] glued-hull-last-dec[OF  $\langle w \in \mathcal{C} \rangle$ ] by blast
  show  $\{u \in \langle \mathcal{C} \rangle. u = \varepsilon \vee \text{last } (\text{Dec } \mathcal{C} \ u) \neq w\} \subseteq \langle \text{glued-gens } w \mathcal{C} \rangle$ 
    using in-glued-hullI by blast
qed

```

```

lemma in-glued-hull-iff:
  assumes  $w \in \mathcal{C}$  and  $u \in \langle \mathcal{C} \rangle$ 
  shows  $u \in \langle \text{glued-gens } w \mathcal{C} \rangle \iff u = \varepsilon \vee \text{last } (\text{Dec } \mathcal{C} \ u) \neq w$ 
  by (simp add:  $\langle w \in \mathcal{C} \rangle$   $\langle u \in \langle \mathcal{C} \rangle \rangle$  code-glued-hull-conv)

```

```

lemma glued-not-in-glued-hull:  $w \in \mathcal{C} \implies w \notin \langle \text{glued-gens } w \mathcal{C} \rangle$ 
  unfolding in-glued-hull-iff[OF - gen-in] code-el-dec
  by (simp add: nemp)

```

```

lemma glued-gens-nemp: assumes  $u \in \text{glued-gens } w \mathcal{C}$  shows  $u \neq \varepsilon$ 
  using assms by (induction) (auto simp add: nemp)

```

```

lemma glued-gens-code: assumes  $w \in \mathcal{C}$  shows code (glued-gens  $w \mathcal{C}$ )
proof

```

```

show  $us = vs$  if  $us \in \text{lists } (\text{glued-gens } w \ C)$  and  $vs \in \text{lists } (\text{glued-gens } w \ C)$ 
and  $\text{concat } us = \text{concat } vs$  for  $us \ vs$ 
using that proof (induction rule: list-induct2')
case ( $\_4 \ u \ us \ v \ vs$ )
  have  $*$ :  $us \in \text{lists } (\text{glued-gens } w \ C) \implies us \in \text{lists } \langle C \rangle$  for  $us$ 
    using sub-lists-mono[OF subset-trans[OF genset-sub glued-hull-sub-hull[OF
 $\langle w \in C \rangle$ ]]].
  obtain  $k \ u' \ l \ v'$ 
    where  $u' \in C \ u' \neq w \ w \ @ \ k \cdot u' = u$ 
    and  $v' \in C \ v' \neq w \ w \ @ \ l \cdot v' = v$ 
    using 4.premis(1-2) by simp (elim conjE in-glued-gensE)
  from this(3, 6) 4.premis  $\langle w \in C \rangle$ 
  have  $\text{concat } (([w] \ @ \ k \cdot [u']) \cdot (\text{Ref } C \ us)) = \text{concat } (([w] \ @ \ l \cdot [v']) \cdot (\text{Ref } C \ vs))$ 
    by (simp add: concat-ref * lassoc)
  with  $\langle w \in C \rangle \langle u' \in C \rangle \langle v' \in C \rangle$  4.premis(1-2)
  have  $[w] \ @ \ k \cdot [u'] \bowtie [w] \ @ \ l \cdot [v']$ 
    by (elim eqd-comp[OF is-code, rotated 2])
    (simp-all add: * pow-in-lists ref-in')
  with  $\langle u' \neq w \rangle \langle v' \neq w \rangle \langle w \ @ \ k \cdot u' = u \rangle \langle w \ @ \ l \cdot v' = v \rangle$ 
  have  $u = v$ 
    by (elim sing-pref-comp-mismatch[rotated 2, elim-format]) blast+
  then show  $u \# us = v \# vs$ 
    using 4.IH 4.premis(1-3) by simp
qed (auto dest!: glued-gens-nemp)
qed

```

A crucial lemma showing the relation between gluing and the decomposition into generators

```

lemma dec-glued-gens: assumes  $w \in C$  and  $u \in \langle \text{glued-gens } w \ C \rangle$ 
shows  $\text{Dec } (\text{glued-gens } w \ C) \ u = \text{glue } w \ (\text{Dec } C \ u)$ 
using  $\langle u \in \langle \text{glued-gens } w \ C \rangle \rangle$  glued-hull-sub-hull'[OF  $\langle w \in C \rangle \langle u \in \langle \text{glued-gens } w \ C \rangle \rangle$ ]
by (intro code.code-unique-dec glued-gens-code)
    (simp-all add: in-glued-hull-iff  $\langle w \in C \rangle$ )

```

```

lemma ref-glue:  $us = \varepsilon \vee \text{last } us \neq w \implies us \in \text{lists } C \implies \text{Ref } C \ (\text{glue } w \ us) = us$ 
by (intro refl glue-in-lists-hull) simp-all

```

**end**

**theorem** glued-code-right:

```

assumes code  $C$  and  $w \in C$ 
shows code  $\{w \ @ \ k \cdot u \mid k \ u. \ u \in C \wedge u \neq w\}$ 
using code.glued-gens-code[OF  $\langle \text{code } C \rangle \langle w \in C \rangle$ ] unfolding glued-gens-alt-def.

```

**theorem** glued-code:

```

assumes code  $C$  and  $w \in C$ 
shows code  $\{u \cdot w \ @ \ k \mid k \ u. \ u \in C \wedge u \neq w\}$ 
using glued-code-right[reversed, OF assms].

```

## 1.6 Gluing is primitivity preserving

It is easy to obtain that gluing lists of code elements preserves primitivity. We provide the result under weaker condition where glue blocks of the list have unique concatenation.

**lemma** (in *code*) *code-prim-glue*:

**assumes** *last-neq*:  $us = \varepsilon \vee \text{last } us \neq w$

**and**  $us \in \text{lists } \mathcal{C}$

**shows**  $\text{primitive } us \implies \text{primitive } (\text{glue } w \ us)$

**using** *prim-map-prim*[*OF prim-concat-prim*, of *decompose C glue w us*]

**unfolding** *refine-def*[*symmetric*] *ref-glue*[*OF assms*].

— In the context of code the inverse to the glue function is the *refine* function, i.e.  $\lambda vs. \text{concat } (\text{map } (\text{decompose } \mathcal{C}) \ vs)$ , see  $\llbracket \text{code } ?\mathcal{C}; ?us = \varepsilon \vee \text{last } ?us \neq ?w; ?us \in \text{lists } ?\mathcal{C} \rrbracket \implies \text{Ref } ?\mathcal{C} \ \text{glue } ?w \ ?us = ?us$ . The role of the *decompose* function outside the code context supply the 'unglue' function, which maps glued blocks to its unique preimages (see below).

**definition** *glue-block* ::  $'a \ \text{list} \Rightarrow 'a \ \text{list} \ \text{list} \Rightarrow 'a \ \text{list} \ \text{list} \Rightarrow \text{bool}$

**where** *glue-block*  $w \ us \ bs =$

$(\exists ps \ k \ u \ ss. (ps = \varepsilon \vee \text{last } ps \neq w) \wedge u \neq w \wedge ps \cdot [w]^{\textcircled{a}} k \cdot u \# ss = us \wedge [w]^{\textcircled{a}} k \cdot [u] = bs)$

**lemma** *glue-blockI* [*intro*]:

$ps = \varepsilon \vee \text{last } ps \neq w \implies u \neq w \implies ps \cdot [w]^{\textcircled{a}} k \cdot u \# ss = us \implies [w]^{\textcircled{a}} k \cdot [u] = bs$

$\implies \text{glue-block } w \ us \ bs$

**unfolding** *glue-block-def* **by** (*intro exI conjI*)

**lemma** *glue-blockE*:

**assumes** *glue-block*  $w \ us \ bs$

**obtains**  $ps \ k \ u \ ss$  **where**  $ps = \varepsilon \vee \text{last } ps \neq w$  **and**  $u \neq w$   $ps \cdot [w]^{\textcircled{a}} k \cdot u \# ss = us$

**and**  $[w]^{\textcircled{a}} k \cdot [u] = bs$

**using** *assms* **unfolding** *glue-block-def* **by** (*elim exE conjE*)

**lemma** *assumes glue-block w us bs*

**shows** *glue-block-of-appendL*: *glue-block*  $w \ (us \cdot vs) \ bs$

**and** *glue-block-of-appendR*:  $vs = \varepsilon \vee \text{last } vs \neq w \implies \text{glue-block } w \ (vs \cdot us) \ bs$

**using**  $\langle \text{glue-block } w \ us \ bs \rangle$  **by** (*elim glue-blockE*, *use nothing in*  $\langle$

*intro glue-blockI*[*of - w - - - vs us \cdot vs bs*]

*glue-blockI*[*OF append-last-neq*, of  $vs \ w \ - \ - \ - \ vs \cdot us \ bs$ ],

*simp-all only: eq-commute*[*of - us*] *rassoc append-Cons refl not-False-eq-True*) $\rangle +$

**lemma** *glue-block-of-block-append*:

$u \neq w \implies \text{glue-block } w \ us \ bs \implies \text{glue-block } w \ ([w]^{\textcircled{a}} k \cdot u \# us) \ bs$

**by** (*simp only: hd-word*[*of - us*] *lassoc*) (*elim glue-block-of-appendR, simp-all*)

**lemma** *in-set-glueE*:

**assumes** *last-neq*:  $us = \varepsilon \vee \text{last } us \neq w$   
**and**  $b \in \text{set } (\text{glue } w \text{ us})$   
**obtains** *bs* **where** *glue-block*  $w \text{ us } bs$  **and**  $\text{concat } bs = b$   
**using** *assms* **proof** (*induction* *us* *rule*: *last-neq-block-induct*)  
**case** (*block*  $k \text{ u } us$ )  
**show** *thesis* **using**  $\langle b \in \text{set } (\text{glue } w ([w]^\text{@ } k \cdot u \# us)) \rangle$   
**proof** (*auto simp add*: *glue-block-append*  $\langle u \neq w \rangle$ )  
**show**  $b = w^\text{@ } k \cdot u \implies \text{thesis}$   
**by** (*auto intro!*: *block.prem*s(1) *glue-blockI*[*OF* -  $\langle u \neq w \rangle$  - *refl*])  
**show**  $b \in \text{set } (\text{glue } w \text{ us}) \implies \text{thesis}$   
**by** (*auto intro!*: *block.IH*[*OF* *block.prem*s(1)] *glue-block-of-block-append*  $\langle u \neq w \rangle$ )  
**qed**  
**qed** *simp*

**definition** *unglue* :: '*a* list  $\Rightarrow$  '*a* list list  $\Rightarrow$  '*a* list  $\Rightarrow$  '*a* list list

**where** *unglue*  $w \text{ us } b = (\text{THE } bs. \text{glue-block } w \text{ us } bs \wedge \text{concat } bs = b)$

**lemma** *unglueI*:

**assumes** *unique-blocks*:  $\bigwedge bs_1 bs_2. \text{glue-block } w \text{ us } bs_1 \implies \text{glue-block } w \text{ us } bs_2$   
 $\implies \text{concat } bs_1 = \text{concat } bs_2 \implies bs_1 = bs_2$   
**shows**  $\text{glue-block } w \text{ us } bs \implies \text{concat } bs = b \implies \text{unglue } w \text{ us } b = bs$   
**unfolding** *unglue-def* **by** (*blast intro*: *unique-blocks*)

**lemma** *concat-map-unglue-glue*:

**assumes** *last-neq*:  $us = \varepsilon \vee \text{last } us \neq w$   
**and** *unique-blocks*:  $\bigwedge vs_1 vs_2. \text{glue-block } w \text{ us } vs_1 \implies \text{glue-block } w \text{ us } vs_2$   
 $\implies \text{concat } vs_1 = \text{concat } vs_2 \implies vs_1 = vs_2$   
**shows**  $\text{concat } (\text{map } (\text{unglue } w \text{ us}) (\text{glue } w \text{ us})) = us$   
**using** *assms* **proof** (*induction* *us* *rule*: *last-neq-block-induct*)  
**case** (*block*  $k \text{ u } us$ )  
**have** *IH*:  $\text{concat } (\text{map } (\text{unglue } w \text{ us}) (\text{glue } w \text{ us})) = us$   
**using** *block.IH*[*OF* *block.prem*s] **by** (*blast intro!*: *glue-block-of-block-append*  $\langle u \neq w \rangle$ )  
**have** \*:  $\text{map } (\text{unglue } w ([w]^\text{@ } k \cdot u \# us)) (\text{glue } w \text{ us}) = \text{map } (\text{unglue } w \text{ us}) (\text{glue } w \text{ us})$   
**by** (*auto simp only*: *map-eq-conv* *unglue-def* *del*: *the-equality* *elim!*: *in-set-glueE*[*OF*  $\langle us = \varepsilon \vee \text{last } us \neq w \rangle$ , *intro the-equality*] (*simp-all only*: *the-equality* *block.prem*s *glue-block-of-block-append*[*OF*  $\langle u \neq w \rangle$ ])  
**show**  $\text{concat } (\text{map } (\text{unglue } w ([w]^\text{@ } k \cdot u \# us)) (\text{glue } w ([w]^\text{@ } k \cdot u \# us))) = [w]^\text{@ } k \cdot u \# us$   
**by** (*auto simp add*: *glue-block-append*[*OF*  $\langle u \neq w \rangle$ ] \* *IH* *intro!*: *unglueI* *intro*: *glue-blockI*[*OF* -  $\langle u \neq w \rangle$ ] *block.prem*s)  
**qed** *simp*

**lemma** *prim-glue*:

**assumes** *last-neq*:  $us = \varepsilon \vee \text{last } us \neq w$

**and** *unique-blocks*:  $\bigwedge bs_1\ bs_2. \text{ glue-block } w\ us\ bs_1 \implies \text{ glue-block } w\ us\ bs_2$   
 $\implies \text{ concat } bs_1 = \text{ concat } bs_2 \implies bs_1 = bs_2$   
**shows** *primitive*  $us \implies \text{ primitive } (\text{ glue } w\ us)$   
**using** *prim-map-prim*[*OF prim-concat-prim, of unglue*  $w\ us\ \text{ glue } w\ us$ ]  
**by** (*simp only: concat-map-unglue-glue assms*)

### 1.6.1 Gluing on binary alphabet

**lemma** *bin-glue-blockE*:  
**assumes**  $us \in \text{ lists } \{x, y\}$   
**and** *glue-block*  $x\ us\ bs$   
**obtains**  $k$  **where**  $[x]^{\textcircled{a}} k \cdot [y] = bs$   
**using** *assms* **by** (*auto simp only: glue-block-def del: in-listsD*)

**lemma** *unique-bin-glue-blocks*:  
**assumes**  $us \in \text{ lists } \{x, y\}$  **and**  $x \neq \varepsilon$   
**shows** *glue-block*  $x\ us\ bs_1 \implies \text{ glue-block } x\ us\ bs_2 \implies \text{ concat } bs_1 = \text{ concat } bs_2$   
 $\implies bs_1 = bs_2$   
**by** (*auto simp: eq-pow-exp*[*OF*  $\langle x \neq \varepsilon \rangle$ ] *elim!*: *bin-glue-blockE*[*OF*  $\langle us \in \text{ lists } \{x, y\} \rangle$ ])

**lemma** *prim-bin-glue*:  
**assumes**  $us \in \text{ lists } \{x, y\}$  **and**  $x \neq \varepsilon$   
**and**  $us = \varepsilon \vee \text{ last } us \neq x$   
**shows** *primitive*  $us \implies \text{ primitive } (\text{ glue } x\ us)$   
**using** *prim-glue*[*OF*  $\langle us = \varepsilon \vee \text{ last } us \neq x \rangle$  *unique-bin-glue-blocks*[*OF assms*( $1-2$ )]]).

**end**

**theory** *Graph-Lemma*  
**imports** *Combinatorics-Words.Submonoids Glued-Codes*

**begin**

## Chapter 2

# Graph Lemma

The Graph Lemma is an important tool for gaining information about systems of word equations. It yields an upper bound on the rank of the solution, that is, on the number of factors into all images of unknowns can be factorized. The most straightforward application is showing that a system of equations admits periodic solutions only, which in particular holds for any nontrivial equation over two words.

The name refers to a graph whose vertices are the unknowns of the system, and edges connect front letters of the left- and right- hand sides of equations. The bound mentioned above is then the number of connected components of the graph.

We formalize the algebraic proof from [1]. Key ingredients of the proof are in the theory *Combinatorics-Words-Graph-Lemma.Glued-Codes*

### 2.1 Graph lemma

**theorem** *graph-lemma-last*:  $\mathfrak{B}_F G = \{\text{last } (\text{Dec } (\mathfrak{B}_F G) g) \mid g. g \in G \wedge g \neq \varepsilon\}$

**proof**

**interpret** *code*  $\mathfrak{B}_F G$

**using** *free-basis-code*.

— the core is to show that each element of the free basis must be a last of some word

**show**  $\mathfrak{B}_F G \subseteq \{\text{last } (\text{Dec } \mathfrak{B}_F G g) \mid g. g \in G \wedge g \neq \varepsilon\}$

**proof** (*rule ccontr*)

— Assume the contrary.

**assume**  $\neg \mathfrak{B}_F G \subseteq \{\text{last } (\text{Dec } \mathfrak{B}_F G g) \mid g. g \in G \wedge g \neq \varepsilon\}$

— And let  $w$  be the not-last

**then obtain**  $w$

**where**  $w \in \mathfrak{B}_F G$

**and** *hd-dec-neg*:  $\bigwedge g. g \in G \implies g \neq \varepsilon \implies \text{last } (\text{Dec } (\mathfrak{B}_F G) g) \neq w$

**by** *blast*

— For contradiction: We have a free hull which does not contain  $w$  but contains

G.

```

have  $G \subseteq \langle \text{glued-gens } w (\mathfrak{B}_F G) \rangle$ 
  by (blast intro! gen-in-free-hull hd-dec-neq del: notI)
then have  $\langle \mathfrak{B}_F G \rangle \subseteq \langle \text{glued-gens } w (\mathfrak{B}_F G) \rangle$ 
  unfolding basis-gen-hull-free
  by (intro code.free-hull-min glued-gens-code  $\langle w \in \mathfrak{B}_F G \rangle$ )
then show False
  using  $\langle w \in \mathfrak{B}_F G \rangle$  glued-not-in-glued-hull by blast
qed
— The opposite inclusion is easy
show  $\{ \text{last } (\text{Dec } \mathfrak{B}_F G g) \mid g. g \in G \wedge g \neq \varepsilon \} \subseteq \mathfrak{B}_F G$ 
  by (auto intro! dec-in-lists lists-hd-in-set[reversed] gen-in-free-hull del: notI)
qed

theorem graph-lemma:  $\mathfrak{B}_F G = \{ \text{hd } (\text{Dec } (\mathfrak{B}_F G) g) \mid g. g \in G \wedge g \neq \varepsilon \}$ 
proof —
  have *:  $\text{rev } u = \text{last } (\text{Dec rev } (\mathfrak{B}_F G) (\text{rev } g)) \wedge g \in G \wedge g \neq \varepsilon$ 
     $\longleftrightarrow u = \text{hd } (\text{Dec } (\mathfrak{B}_F G) g) \wedge g \in G \wedge g \neq \varepsilon$  for  $u g$ 
  by (cases  $g \in G \wedge g \neq \varepsilon$  (simp add: gen-in-free-hull last-rev hd-map code.dec-rev, blast))
  show ?thesis
  using graph-lemma-last[reversed, of G] unfolding *.
qed

```

## 2.2 Binary code

We illustrate the use of the Graph Lemma in an alternative proof of the fact that two non-commuting words form a code. See also  $\llbracket u_0 \cdot u_1 \neq u_1 \cdot u_0; us \in \text{lists } \{u_0, u_1\}; vs \in \text{lists } \{u_0, u_1\}; \text{concat } us = \text{concat } vs \rrbracket \implies us = vs$  in *Combinatorics-Words.CoWBasic*.

First, we prove a lemma which is the core of the alternative proof.

```

lemma non-comm-hds-neq: assumes  $u \cdot v \neq v \cdot u$  shows  $\text{hd } (\text{Dec } \mathfrak{B}_F \{u, v\} u) \neq \text{hd } (\text{Dec } \mathfrak{B}_F \{u, v\} v)$ 
using assms proof (rule contrapos-nn)
  assume hds-eq:  $\text{hd } (\text{Dec } \mathfrak{B}_F \{u, v\} u) = \text{hd } (\text{Dec } \mathfrak{B}_F \{u, v\} v)$ 
  have **:  $\mathfrak{B}_F \{u, v\} = \{ \text{hd } (\text{Dec } \mathfrak{B}_F \{u, v\} u) \}$ 
    using graph-lemma by (rule trans) (use assms in  $\langle \text{auto intro: hds-eq[symmetric] \rangle$ )
  show  $u \cdot v = v \cdot u$ 
    by (intro comm-rootI[of - hd (Dec  $\mathfrak{B}_F \{u, v\} u$ )])
    (simp-all add: **[symmetric] gen-in-free-hull)
qed

```

```

theorem assumes  $u \cdot v \neq v \cdot u$  shows code  $\{u, v\}$ 
proof (rule code.intro)
  have *:  $w \in \{u, v\} \implies w \neq \varepsilon$  for  $w$ 
    using  $\langle u \cdot v \neq v \cdot u \rangle$  by blast
  fix  $xs\ ys$ 

```

```

show  $xs \in \text{lists } \{u, v\} \implies ys \in \text{lists } \{u, v\} \implies \text{concat } xs = \text{concat } ys \implies xs =$ 
 $ys$ 
proof (induction xs ys rule: list-induct2')
  case ( $\lambda x xs y ys$ )
    have **:  $\text{hd } (\text{Dec } \mathfrak{B}_F \{u, v\} (\text{concat } (z \# zs))) = \text{hd } (\text{Dec } \mathfrak{B}_F \{u, v\} z)$ 
    if  $z \# zs \in \text{lists } \{u, v\}$  for  $z zs$ 
    using that by (elim listsE) (simp del: insert-iff
      add: concat-in-hull' gen-in set-mp[OF hull-sub-free-hull]
      free-basis-dec-morph * basis-gen-hull-free)
    have  $\text{hd } (\text{Dec } \mathfrak{B}_F \{u, v\} x) = \text{hd } (\text{Dec } \mathfrak{B}_F \{u, v\} y)$ 
    using 4.prems by (simp only: **[symmetric])
    then have  $x = y$ 
    using 4.prems(1-2) non-comm-hds-neq[OF <u · v ≠ v · u>]
    by (elim listsE insertE emptyE) simp-all
    with  $\lambda$  show  $x \# xs = y \# ys$  by simp
  qed (simp-all add: *)
qed
end

```



# References

- [1] J. Berstel, D. Perrin, J. Perrot, and A. Restivo. Sur le théorème du défaut. *Journal of Algebra*, 60(1):169–180, 1979.